1 Syntax

1.1 Syntax Notation

This specification uses the Augmented Backus-Naur Form (ABNF) notation of [RFC2234], including the following core ABNF syntax rules defined by that specification: ALPHA (letters), DIGIT (decimal digits), and HEXDIG (hexadecimal digits). The complete XRI syntax is collected in Appendix A. 

1.2 Characters

The character set and encoding of an XRI is inherited from IRI syntax as defined in [IRI].  That is, the URI character set defined in [2396bis] is expanded to include the characters of the UCS (Universal Character Set, [ISO10646]) beyond U+007F. 

1.3 Character Encoding

The basic character encoding of XRI is UTF-8, as recommended by [RFC2718]. When an XRI is presented as a human readable identifier, the representation of the XRI in the underlying document should use the character encoding of the underlying document. However, this string must be converted to UTF-8 before any processing external to the underlying document. 

Note that not all ASCII sequences can be derived from UTF-8 sequences. A valid XRI character sequence MUST be derivable by unescaping an equivalent UTF-8 sequence. For example, the ASCII sequence '%FC', which would represent U+00FC LATIN SMALL LETTER U WITH DIAERESIS in an iso-8859-1 encoding, when unescaped will not result in a valid UTF-8 sequence. [ML: I not convinced that this point is important for us to call out.  Does anyone think we need this text?]

1.3.1 Reserved Characters

XRI reserves the same characters as [IRI].  While the same characters are reserved, a number of characters are taken from the sum-delims category of the [IRI] syntax and are used as delimiters in the XRI syntax (xri-gen-delims).


xri-reserved      = xri-gen-delims / xri-sub-delims

  xri-gen-delims    = ":" / "/" / "?" / "#" / "[" / "]" / "("

                    / ")" / "*" / gcs-char

  xri-sub-delims    = "&" / ";" / "," / "'"

  gcs-char          = "=" / "@" / "+" / "$" / "!"

If the use of an unescaped XRI reserved character as a data character would cause the interpretation of the XRI to be ambiguous, the character MUST be escaped as per the rules in section 1.3.3, “Escaped Characters”, and particularly section 2.1.1.

1.3.2 Unreserved Characters

The characters that are allowed in XRIs and do not have a reserved purpose are called unreserved.  XRI has the same set of unreserved characters as [IRI] (iunreserved). [ML: Just noticed that this is the case – we should probably just remove the xri-unreserved production from the ABNF.]

  iunreserved       = ALPHA / DIGIT / "-" / "." / "_" / "~" / ucschar

  ucschar           = %xA0-D7FF / %xF900-FDCF / %xFDF0-FFEF /

                    %x10000-1FFFD / %x20000-2FFFD / %x30000-3FFFD /

                    %x40000-4FFFD / %x50000-5FFFD / %x60000-6FFFD /

                    %x70000-7FFFD / %x80000-8FFFD / %x90000-9FFFD /

                    %xA0000-AFFFD / %xB0000-BFFFD / %xC0000-CFFFD /

                  %xD0000-DFFFD / %xE1000-EFFFD
Escaping unreserved characters in an XRI does not impact what resource is identified by that XRI. However, it may change the result of an XRI comparison (see section [normalization]) so unreserved characters should not be escaped unless necessary.

1.3.3 Percent-encoded Characters

XRIs follow the same rules for escaping characters as IRIs and URIs. That is, any characters in an XRI MUST be escaped if: a) they are not in the allowed character set, and b) using a reserved character could cause the XRI to be misinterpreted. 

An XRI thus escaped is said to be in escaped normal form. This does not imply that it is necessarily a valid IRI or URI. Rules for converting an XRI into a valid IRI or URI are discussed in section [Converting XRIs to IRIs and URIs]. 

1.3.3.1 Encoding XRI Metadata

In some cases, the transformation from an identifier in its native language and display format into an XRI in escaped normal form may lose information that cannot be retained through character escaping. For example, in certain languages displaying the glyph of a UTF-8 encoded character requires additional language and font information not available in UTF-8. The loss of this information during UTF-8 encoding could cause the resulting XRI to be ambiguous.

Another case is when the normalization or canonicalization rules of a particular identifier authority do not permit the inclusion of mixed case letters, or certain punctuation in an XRI segment even when escaped, yet the authority would like to retain this metadata for purposes of presentation. XRI syntax offers an option for encoding this metadata using a cross-reference beginning with the GCS “$” symbol. As defined in section 1.4.1.3, the top level authority for these identifiers is the “XRI Metadata Specification” Error! Reference source not found.. It defines special identifiers for UTF-8 metadata, presentation metadata, and other standard types of identifier metadata together with the rules governing their interpretation. [ML: need to clean up the preceding paragraph.]
1.4 Syntax Components

Generic XRI syntax builds on generic IRI (and ultimately, URI) syntax. However because XRI syntax includes syntactic elements other than those defined in [IRI] and [2396bis], this specification does not define a new URI scheme. Instead, it follows the example of [IRI] and defines a new identifier scheme, along with a specification for transforming XRIs into generic URIs or IRIs for applications that expect them (see section [transformation]).

As with URIs, an XRI may be either absolute or relative.


XRI-reference     = XRI / relative-XRI

  XRI               = [ "xri://" ] xri-hier-part [ "?" xri-query ] 

                    [ "#" xri-fragment ]


absolute-XRI      = [ "xri://" ] xri-hier-part [ "?" xri-query ]

  relative-XRI      = xri-path [ "?" xri-query ] [ "#" xri-fragment ]

  xri-hier-part     = ( xri-authority / iauthority ) 

                      [ xri-path-absolute ] / ipath-empty

An absolute XRI consists of the scheme name “xri:” followed by the same set of hierarchical components as an absolute URI – authority, path, and query. A relative XRI consists of either a local path or a relative path.

Finally, in certain contexts such as cross-references (section 1.4.1.4), the “xri:” scheme name is redundant. These contexts can use the xri-value production, which includes all levels of XRI paths.

1.4.1 Authority

XRI syntax supports the same types of authorities as generic IRI syntax, called IRI authorities. In addition, it supports XRI authorities that provide two other mechanisms for specifying the global context of an identifier, as defined in section 2.1.1.2.

1.4.1.1 
XRI Authority

In addition to the authorities supported in generic URI syntax, XRIs support two other mechanisms for specifying the global context of an identifier. The first is the global context symbol (GCS), and the second is the cross-reference (abbreviated in the ABNF as xref).


xri-authority     = xref-authority / gcs-authority

  xref-authority    = xref *xri-subseg

  gcs-authority     = gcs-char xri-segment

1.4.1.2 IRI Authority

In the context of an XRI, a URI authority is distinguished by an initial double slash (“//”). 


  iauthority        = [ iuserinfo "@" ] ihost [ ":" port ]

The syntax following this starting delimiter is inherited directly from Error! Reference source not found., which simplifies the syntax in Error! Reference source not found. and includes support for IPv6 addresses defined in Error! Reference source not found.. First, the “userinfo” sub-component permits identifying a user in the context of a host.

Next, the “host” sub-component has three options for identifying the host: a domain name, an IPv4 address, or an IPv6 literal. 

Note that the host identifier may be omitted. This is because in generic URI syntax, a default may be defined by the semantics of a particular URI scheme. No default is specified for the XRI scheme; this allows a default to be inherited from the particular protocol used to resolve the XRI. 

A hostname, after the transformation described in step 4 of section Error! Reference source not found., MUST meet the rules defined in section 3.2.2 of Error! Reference source not found.. The productions for idomainlabel, qualified, and hostname, therefore, have additional restrictions not reflected in the ABNF. See section Error! Reference source not found..
Support for an IPv6 address literal was added by Error! Reference source not found. following the syntax originally specified in Error! Reference source not found.. Because IPv6 literals use colons as delimiters, they must be encapsulated within square brackets.

Finally, a host identifier can be followed by an optional port number. Because XRIs are abstract identifiers, the XRI syntax specification does not define a default port. It is expected that the default port will be inherited from the resolution protocol, such as the HTTP/HTTPS protocol specified in section Error! Reference source not found.. Therefore, if the port is omitted in an XRI, it is undefined.

1.4.1.3 Global Context Symbols (GCS)

To support the abstraction and human-friendly identifier (HFI) requirements, XRIs offer a simple, compact syntax for indicating the logical global context of an identifier: a single prefix character.


 gcs-char          = "=" / "@" / "+" / "$" / "!" 

The global context symbol characters were selected from the set of symbol characters that are valid in a URI under Error! Reference source not found. to represent the global contexts shown in Table 1:


	Symbol Character
	Authority Type
	Establishes global context for

	+
	General public
	Identifiers for generic concepts for which there is no specific authority, i.e., that are established by public convention. (In the English language, for example, these would be the generic nouns.)

	=
	Person
	Identifiers that represent an individual person.

	@
	Organization
	Identifiers that represent an organization of any kind.

	$
	OASIS XRI Metadata
Specification
	Special identifiers established by the “XRI Metadata Specification” Error! Reference source not found. for interoperable identifier metadata (e.g., language, version, type, query syntax, etc.).

	*
	User-relative
	Identifiers for which the authority is relative to the current user (“shortcut XRIs”). See section Error! Reference source not found..

	!
	XRI author
	Identifiers used only for human-readable annotations of XRIs (ignored by machine processing.)


Table 1: XRI global context symbols.

Note that because a global context symbol may precede an xri-segment, and an xri-segment may start with a cross-reference (below), a global context symbol can be used to express the abstract logical context of a conventional URI authority. For example:



xri:=(http://www.my-website.com)/favorites.html



--expresses that this resource represents an individual

1.4.1.4 Cross-References

Cross-references are the primary extensibility mechanism in XRIs. They allow an identifier assigned in one context to be reused in another context, permitting identifiers to be shared across contexts to simplify identifying logically equivalent resources. To syntactically delimit a cross-reference, it is enclosed in parentheses the same way an IPv6 literal is encapsulated in square brackets as specified in Error! Reference source not found. (see section 2.1.1.1). A cross-reference may contain either an XRI value or an absolute URI.


  xref              = "(" ( XRI-reference / IRI ) ")"

It is important that the value of a cross-reference be syntactically unambiguous, whether it is an absolute URI or one of the various forms of an XRI value. Therefore special attention must be paid to relative XRIs to avoid ambiguity, as discussed in 2.3.3. 

A cross-reference may appear at any node of any XRI except within a URI authority segment. The use of cross-references as the very first segment in an XRI enables any globally-unique identifier in any URI scheme (e.g., an HTTP URI, mailto URI, URN, etc.) to specify a global authority.



xri:(mailto:john.doe@example.com)/favorites/home



--example of using a URI as an XRI global authority

1.4.1.5 Self-References

Cross-reference syntax is also the means by which an XRI can express that it is not intended for resolution, but only for the purpose of establishing equivalence across contexts. Such an XRI is called a self-reference. To express a self-reference, the entire XRI value is enclosed in parentheses—in essence, it becomes a global cross-reference. This is the XRI equivalent of the English language convention of putting a word or phrase in quotes to express that the author is referring to the word or phrase itself and not to its normal meaning. (In linguistics and philosophy, this is called the “use-mention distinction.”) For example:


The term "user-friendly" is used frequently in computing.



--English-language usage of a quoted term


xri:(+user-friendly)



--XRI syntax for expressing a self-reference

1.4.2 Path

As with URIs, the XRI path component is a hierarchal sequence of path segments separated by slash (“/”) characters and terminated by the first question-mark (“?”) or number sign (“#”) character, or by the end of the XRI. The key difference is that while a URI path segment is considered opaque by a generic URI processor, an XRI path segment can be parsed by an XRI processor into two types of sub-segments: dot segments and colon segments after their leading characters (“.” and “:”).


xri-path          = xri-path-absolute

                    / xri-path-noscheme

                    / ipath-empty  

  xri-path-absolute = "/" [ xri-segment-nz *( "/" xri-segment ) ]

  xri-path-noscheme = xri-subseg-od-nx *xri-subseg-nc *( "/" xri-segment )

  xri-segment       = xri-subseg-od *xri-subseg

  xri-segment-nz    = xri-subseg-od-nz *xri-subseg

  xri-subseg        = ( "*" / "!" ) (xref / *xri-pchar)

  xri-subseg-nc     = ( "*" / "!" ) (xref / *xri-pchar-nc)

  xri-subseg-od     = [ "*" / "!" ] (xref / *xri-pchar)

  xri-subseg-od-nz  = [ "*" / "!" ] (xref / 1*xri-pchar)

  xri-subseg-od-nx  = [ "*" / "!" ] 1*xri-pchar-nc

Dot segments are used to specify reassignable identifiers—identifiers that may be reassigned by an identifier authority to represent a different resource at some future date. Colon segments (following the lead of URN syntax in Error! Reference source not found.) are used to specify persistent identifiers—identifiers that are permanently assigned to a resource and will not be reassigned at a future date. The default is a dot segment, so no leading dot is required if this is the first (or only) sub-segment.

Note that for compatibility with URI syntax, the ABNF allows two special values of an XRI segment—a single dot and a double dot. These can be used as the leading characters in a relative XRI to indicate its relationship to a base XRI as specified in Error! Reference source not found..
Other than these special uses of the dot (“.”) and the colon (“:”) characters, an XRI path segment can contain the same characters as a URI path segment plus the expanded UCS character set (section 1.3.2). If a dot or colon is used, it will be interpreted as a delimiter. If this interpretation is not desired for these characters, or for any other special XRI delimiters, these characters MUST be escaped when they appear in the path segment. See section 1.3.3, “Percent-encoded Characters”.


xri-pchar         = xri-unreserved / pct-encoded / xri-sub-delims 

                    / ":"

  xri-pchar-nc      = xri-unreserved / pct-encoded / xri-sub-delims

With the exception of dot and colon sub-segments, an XRI path segment is considered opaque by generic XRI syntax. As with URIs in general, XRI extensions or generating applications may define special meanings for other URI reserved characters for the purpose of delimiting extension-specific or generator-specific sub-components. For example, section 3.4 of Error! Reference source not found. specifies the set of URI reserved characters that can be used within a query segment. 

1.4.3 Query

The XRI query component is identical to the URI query component as described in section 3.4 of Error! Reference source not found., except that it allows the full XRI character range and it may begin with a cross-reference. The latter feature permits the incorporation of XRI metadata describing the query string syntax. See the “XRI Metadata Specification” Error! Reference source not found. for more about query syntax metadata.


  xri-query         = *( xri-pchar / iprivate / "/" / "?"

                    / "@" / "(" / ")" / "!" / "$" / "*" / "+" / "=" )

1.4.4 Fragment

XRI syntax also supports fragments as described in section 4.1 of Error! Reference source not found., except that it allows the full XRI character range and may begin with a cross-reference. 


  xri-fragment      = *( xri-pchar / "/" / "?"

                    / "@" / "(" / ")" / "!" / "$" / "*" / "+" / "=" )

Since XRI syntax can directly address attributes or secondary representations of a primary resource to any depth, fragments are supported primarily for compatibility with generic URI syntax. XRIs can also employ cross-references to identify media types or other alternative representations of a resource. See section 

Appendix A. Transforming HTTP URIs to XRIs (Non-Normative)

To leverage existing infrastructure, it may sometimes be useful to convert HTTP URIs into XRIs. Because XRI syntax is, for the most part, a superset of generic URI syntax, the majority of HTTP URIs can be converted to valid XRIs simply by replacing the scheme “http” with “xri”. Special consideration, however, must be given to HTTP URIs employing the characters in the “xri-reserved” production of this specification that differ from those in the “reserved” production of Error! Reference source not found. (as amended by Error! Reference source not found.). These include opening parenthesis (“(“), closing parenthesis (“)”), dot (“.”), asterisk (“*”), and exclamation point (“!”). 

Typically, characters in the “reserved” production of Error! Reference source not found. that appear in an HTTP URI as normal characters (i.e. not as syntactic delimiters) are escaped encoded. However, this is not required in all cases. Error! Reference source not found. says

“Characters in the ‘reserved’ set are not reserved in all contexts. The set of characters actually reserved within any given URI component is defined by that component. In general, a character is reserved if the semantics of the URI changes if the character is replaced with its escaped US-ASCII encoding.”

Characters in the “xri-reserved” set that are properly left un-escaped in an HTTP URI may be semantically significant when the HTTP URI is converted to an XRI. For example,

http://www.example.com/example1:example2

is a valid HTTP URI even though it contains an unescaped reserved character – a colon (“:”) – because section 3.3 of Error! Reference source not found. explicitly omits this character from the reserved set for “path” components. The same unescaped character in an XRI, however, will be interpreted as a delimiter. If the colon character should not be understood as a delimiter in the resulting XRI, it must be escaped during conversion. The same applies to the other characters mentioned above.

Generally, any character not in the “xri-pchar” set that appears in the “abs_path”, “query”, or “fragment” components of the HTTP URI will need to be escaped when converting to an XRI. This avoids misinterpretation in the resulting XRI following the guidance in section 2.2.4 of this specification. 

Exceptions are possible. For example, if the author of the above HTTP URI intended the colon character to be interpreted as described in this specification, or if its use would not be misinterpreted, then it may be left in its unescaped form.

In addition, it may be beneficial to escape other characters like the percent (“%”) character, particularly if it may be necessary to convert the resulting XRI back to an HTTP URI. Whether such additional escaping is desirable or not depends on the intended use of the resulting XRI, the context in which it will appear, how it is intended to be resolved, etc.

It is worth noting that some rare forms of HTTP URIs can result in XRIs that are misleading to the reader. For example, the following unusual HTTP URI is valid per Error! Reference source not found..

http://@example.com/example1

When converted to an XRI, as

xri://@example.com/example1

a casual reader could easily misinterpret the “uri-authority” component as an “xri-authority”. Similarly, a URI with an authority segment like

http://=bob@example.com/example1

could be similarly misinterpreted.

